

Lesson 21

Triple Integrals in Cylindrical and Spherical Coordinates

Initializations

```
> restart;  
with(plots):  
with(student):  
>
```

21.1 Triple Integrals in Cylindrical Coordinates

Recall that cylindrical coordinates are defined by the equations:

$$\begin{aligned}x &= r \cos \theta \\y &= r \sin \theta \\z &= z\end{aligned}$$

Transformation of triple integrals into cylindrical coordinates is accomplished by

$$dV = \left| \frac{\partial(x, y, z)}{\partial(r, \theta, z)} \right| dr d\theta dz = r dr d\theta dz$$

Examples

Example 21.1.1

Evaluate

$$\iiint_E x^2 z dV$$

where E denotes the region above the xy plane bounded by the planes $z = 0$ and $z = y$, and the cylinder $x^2 + y^2 = 1$.

Solution

The floor of the region E is the half circle $x^2 + y^2 = 1, y \geq 0$, in the xy -plane and the ceiling is given by $z = y$. A nice picture can be created, but as always it pays to make a rough sketch before attempting to create the plot syntax. Because of the circular nature of the problem, it is easiest to use cylindrical coordinates, both for the plotting and for the actual computation. In the picture below the cylinder has been opened up a bit to allow for a glance inside the solid.


```
> cyl:={x=r*cos(theta), y=r*sin(theta), z=z};  
cyl:= {x=r*cos(theta), y=r*sin(theta), z=z}
```

(2.1.1.1)

```

> p11:=plot3d(subs(r=1, subs(cyl, [x, y, z])), theta=1.2.
.2*Pi, z=0..1.1, color=green):
p12:=plot3d(subs(r=1, subs(cyl, [x, y, z])), theta=0.
.0.6, z=0..1.1, color=green, grid=[6, 25]):
p2:=plot3d(y, x=-1.1..1.1, y=0..1.1, color=red, grid=[15,
15]):
p3:=plot3d(0, x=-1.1...1.1, y=0..1.1, color=grey, grid=
[15,15]):
display([p11, p12, p2, p3], view=[-1.1..1.1, 0..1.1, 0.
.1.1], axes=boxed, labels=[x, y, z], orientation=[49, 68]
, scaling=constrained);

```



We plot the footprint of this solid in the xy -plane.

```

> f:=(x, y, z)->x^2*z;
bnd_1:=x^2+y^2=1;
bnd_2:=y=0;

```

$$f := (x, y, z) \rightarrow x^2 z$$

$$bnd_1 := x^2 + y^2 = 1$$


$$bnd_2 := y = 0$$

(2.1.1.2)

```

> implicitplot({bnd_1, bnd_2}, x=-1.2..1.2, y=-0.2..1.2,
axes=boxed, scaling=constrained);

```



We first integrate over z , then over r , and finally over θ . In order to increase transparency, we assign variable names to the limits for z , r , and θ .

```
> zmin:=0;
   zmax:=subs(cyl, y);
   rmin:=0;
   rmax:=1;
   thetamin:=0;
   thetamax:=Pi;
```

$$\begin{aligned}
 zmin &:= 0 \\
 zmax &:= r \sin(\theta) \\
 rmin &:= 0 \\
 rmax &:= 1 \\
 thetamin &:= 0 \\
 thetamax &:= \pi
 \end{aligned} \tag{2.1.1.3}$$

Now setup and evaluate the integral.

```
> e1:=Tripleint(subs(cyl, f(x, y, z))*r, z=zmin..zmax, r=
   rmin..rmax, theta=thetamin..thetamax);
```

$$e1 := \int_0^{\pi} \int_0^1 \int_0^{r \sin(\theta)} r^3 \cos(\theta)^2 z \, dz \, dr \, d\theta \tag{2.1.1.4}$$

```
> answer:=value(e1);
   evalf(%);
```

$$\begin{aligned}
 answer &:= \frac{1}{96} \pi \\
 &0.03272492349
 \end{aligned} \tag{2.1.1.5}$$

▼ 21.2 Triple Integrals in Spherical Coordinates

Recall that spherical coordinates are defined by the equations:

$$x = \rho \cos \theta \sin \phi$$

$$y = \rho \sin \theta \sin \phi$$

$$z = \rho \cos \phi$$

Transformation of triple integrals into spherical coordinates is achieved by

$$dV = \left| \frac{\partial(x, y, z)}{\partial(\rho, \theta, \phi)} \right| d\rho d\theta d\phi = \rho^2 \sin \phi d\rho d\theta d\phi$$

▼ Examples

▼ Example 21.2.1


Integrate the function $f(x, y, z) = \sqrt{x^2 + y^2 + z^2 + 5}$ over the region in three space bounded by the cone $\phi = \frac{\pi}{3}$ and the sphere $x^2 + y^2 + z^2 = 12$.

Solution

Observe that the given region looks like an ice cream cone. Because of the spherical nature of the problem we use spherical coordinates.

```
> sc:={x=rho*cos(theta)*sin(phi), y=rho*sin(theta)*sin(phi)
, z=rho*cos(phi)};
      sc:={x=ρ cos(θ) sin(φ), y=ρ sin(θ) sin(φ), z=ρ cos(φ)} (3.1.1.1)

> p1:=sphereplot(sqrt(12), theta=Pi/2..2*Pi, phi=0..
Pi/3+0.2, style=patch, color=green):
p2:=plot3d(subs(phi=Pi/3, subs(sc, [x, y, z])), rho=0..
sqrt(12)+0.5, theta=0..2*Pi, style=patch, color=gold):
display([p1, p2], orientation=[50, 66], axes=boxed,
labels=[x, y, z]);
```



Enter the data and setup the integral.

```
> f:=(x, y, z)->sqrt(x^2+y^2+z^2+5);
cone:=phi=Pi/3;
sphere:=x^2+y^2+z^2=12;
```

$$f := (x, y, z) \rightarrow \sqrt{x^2 + y^2 + z^2 + 5}$$

$$\text{cone} := \phi = \frac{1}{3} \pi$$

$$\text{sphere} := x^2 + y^2 + z^2 = 12 \quad (3.1.1.2)$$

```
> e1:=Tripleint(subs(sc, f(x, y, z))*rho^2*sin(phi), rho=0.
.sqrt(12), theta=0..2*Pi, phi=0..Pi/3);
```

$$e1 := \int_0^{2\sqrt{3}} \int_0^{2\pi} \int_0^{\frac{1}{3}\pi} \sqrt{\rho^2 \cos(\theta)^2 \sin(\phi)^2 + \rho^2 \sin(\theta)^2 \sin(\phi)^2 + \rho^2 \cos(\phi)^2 + 5} \rho^2 \sin(\phi) \, d\phi \, d\theta \, d\rho \quad (3.1.1.3)$$

$$\int_0^{2\sqrt{3}} \int_0^{2\pi} \int_0^{\frac{1}{3}\pi} \sqrt{\rho^2 \cos(\theta)^2 \sin(\phi)^2 + \rho^2 \sin(\theta)^2 \sin(\phi)^2 + \rho^2 \cos(\phi)^2 + 5} \rho^2 \sin(\phi) \, d\phi \, d\theta \, d\rho$$

Simplification of this expression will aid the computation.

```
> e2:=simplify(e1);
```

$$e2 := \left(\int_0^{2\sqrt{3}} \sqrt{\rho^2 + 5} \rho^2 \, d\rho \right) \left(\int_0^{2\pi} 1 \, d\theta \right) \left(\int_0^{\frac{1}{3}\pi} \sin(\phi) \, d\phi \right) \quad (3.1.1.4)$$

```
> e3:=simplify(value(e2));
```

$$(3.1.1.5)$$

$$e3 := -\frac{1}{16} \pi (100 \ln(2) + 25 \ln(3) - 25 \ln(5) - 116 \sqrt{17} \sqrt{3} - 50 \ln(12) + 50 \ln(6 + \sqrt{17} \sqrt{3})) \quad (3.1.1.5)$$

This wild expression can be simplified by collecting the radicals and the logarithms.

```
> answer:=combine(e3, [radical, ln]);
evalf(%);
```

$$answer := -\frac{1}{16} \pi \left(-116 \sqrt{51} + 100 \ln \left(\frac{1}{15} \sqrt{3} 375^{1/4} \sqrt{6 + \sqrt{51}} \right) \right) \\ 150.6626287 \quad (3.1.1.6)$$

```
>
```